Math 132
Summer 2008 Exam II

1. Calculate $\displaystyle\int_0^1 \frac{1}{\sqrt{5-4x-x^2}}\,dx$.

Solution:

```
> J := Int(1/sqrt(5-4*x-x^2),x = 0 .. 1);
```

$$J := \int_0^1 \frac{1}{\sqrt{5-4x-x^2}}\,dx$$

```
> p := 5-4*x-x^2;
```

$$p := 5-4x-x^2$$

```
> p := student[completesquare](p,x);
```

$$p := -(x+2)^2+9$$

```
> J1 := subs(5-4*x-x^2 = -(x+2)^2+9,  J);
```

$$J1 := \int_0^1 \frac{1}{\sqrt{-(x+2)^2+9}}\,dx$$

```
> J2 := student[changevar](u = x+2, J1, u);
```

$$J2 := \int_2^3 \frac{1}{\sqrt{-u^2+9}}\,du$$

```
> Int(1/sqrt(a^2-u^2),u) = arcsin(u/a) + C;
```

$$\int \frac{1}{\sqrt{a^2-u^2}}\,du = \arcsin\left(\frac{u}{a}\right)+C$$

```
> Answer := subs({u=3,a=3}, arcsin(u/a) + C ) - subs({u=2,a=3},
  arcsin(u/a) + C );
```

$$Answer := \arcsin(1) - \arcsin\left(\frac{2}{3}\right)$$

```
> Answer := simplify(Answer);
```

$$Answer := \frac{\pi}{2} - \arcsin\left(\frac{2}{3}\right)$$

```
> value(J);  #Verification by allowing Maple to do the integral without help
```

$$\frac{\pi}{2} - \arcsin\left(\frac{2}{3}\right)$$

2. Calculate $\int x \ln(x)\, dx$.

Solution:

```
> J := Int(x*ln(x),x);
```

$$J := \int x \ln(x)\, dx$$

```
> u := ln(x);  #part of integrand to be differentiated; rest of integrand is dv
```

$$u := \ln(x)$$

```
> J1 := student[intparts](J, u);
```

$$J1 := \frac{1}{2} \ln(x)\, x^2 - \int \frac{x}{2}\, dx$$

```
> Answer := value(J1);
```

$$Answer := \frac{1}{2} \ln(x)\, x^2 - \frac{x^2}{4}$$

Remark: **Maple** does not use absolute values in the formula $\int \frac{1}{x}\, dx = \ln(|x|) + C$ because **Maple** employs complex values for its evaluations and can thereby assign values to ln(x) when x < 0.

3. Calculate $\int x^2 e^{(-x)}\, dx$.

Solution:

```
> J := Int(x^2*exp(-x),x);
```

$$J := \int x^2\, \mathbf{e}^{(-x)}\, dx$$

```
> u := x^2;  #part of integrand of J to be differentiated; rest
  of integrand is dv
```

$$u := x^2$$

```
> J1 := student[intparts](J, u);
```

$$J1 := -x^2\, \mathbf{e}^{(-x)} - \int -2\, x\, \mathbf{e}^{(-x)}\, dx$$

```
> u := x;  #part of integrand of J1 to be differentiated; rest of
  integrand is dv
```

$$u := x$$

```
> J2 := student[intparts](J1, u);
```

$$J2 := -x^2\, \mathbf{e}^{(-x)} - 2\, x\, \mathbf{e}^{(-x)} + \int 2\, \mathbf{e}^{(-x)}\, dx$$

```
> Answer := simplify(value(J2));
```

$$Answer := -\mathbf{e}^{(-x)}\, (x^2 + 2\, x + 2)$$

4. Calculate $\int_{\frac{\pi}{6}}^{\frac{\pi}{4}} \sin(x)^2\, dx$.

Solution:

```
> J := Int(sin(x)^2,x = Pi/6 .. Pi/4);
```

$$J := \int_{\frac{\pi}{6}}^{\frac{\pi}{4}} \sin(x)^2\, dx$$

```
> J := Int((1-cos(2*x))/2,x = Pi/6 .. Pi/4);
```

$$J := \int_{\frac{\pi}{6}}^{\frac{\pi}{4}} \frac{1}{2} - \frac{1}{2}\cos(2\,x)\,dx$$

```
> Answer := simplify(value(J));
```

$$Answer := \frac{\pi}{24} - \frac{1}{4} + \frac{\sqrt{3}}{8}$$

5. Calculate $\int \sin(x)^2 \cos(x)^3\,dx$.

Solution:

```
> u := 'u': #Restores u to its unassigned state (Otherwise Maple
  would remember u from a previous            problem and that
  value would interfere with carrying out this problem.)
> J := Int(sin(x)^2*cos(x)^3,x);
```

$$J := \int \sin(x)^2 \cos(x)^3\,dx$$

```
> eqn := cos(x)^3 = (1-sin(x)^2)*cos(x);
```

$$eqn := \cos(x)^3 = (1 - \sin(x)^2)\cos(x)$$

```
> J1 := subs(eqn, J);
```

$$J1 := \int \sin(x)^2 (1 - \sin(x)^2)\cos(x)\,dx$$

```
> J2 := Int(u^2*(1-u^2),u);  #Make change of variable u = sin(x),
  du = cos(x)*dx
```

$$J2 := \int u^2 (1 - u^2)\,du$$

```
> J3 := value(J2) + C;
```

$$J3 := -\frac{u^5}{5} + \frac{u^3}{3} + C$$

```
> Answer := subs(u = sin(x), J3);
```

$$Answer := -\frac{1}{5}\sin(x)^5 + \frac{1}{3}\sin(x)^3 + C$$

```
> value(J);
```

$$-\frac{1}{5}\sin(x)\cos(x)^4+\frac{1}{15}\cos(x)^2\sin(x)+\frac{2}{15}\sin(x)$$

Let's verify this answer:

```
> Answer2 := value(J); #The answer Maple gets unaided
```

$$Answer2 := -\frac{1}{5}\sin(x)\cos(x)^4+\frac{1}{15}\cos(x)^2\sin(x)+\frac{2}{15}\sin(x)$$

```
> testeq(diff(Answer,x) = diff(Answer2,x));
```

true

6. Calculate $\displaystyle\int \frac{1}{\sqrt{4+x^2}}\,dx$**.**

Solution:

```
> J := Int(1/sqrt(4+x^2),x);
```

$$J := \int \frac{1}{\sqrt{4+x^2}}\,dx$$

```
> J1 := student[changevar](x = 2*tan(theta), J, theta);
```

$$J1 := \int \frac{2+2\tan(\theta)^2}{\sqrt{4+4\tan(\theta)^2}}\,d\theta$$

```
> J2 := Int(sec(theta),theta);    # Using 1 + tan(theta)^2 = 
sec(theta)^2 and simplifying
```

$$J2 := \int \sec(\theta)\,d\theta$$

```
> J3 := value(J2) + C;
```

$$J3 := \ln(\sec(\theta)+\tan(\theta))+C$$

```
> Answer := subs({sec(theta) = sqrt(4+x^2)/2, tan(theta) = x/2}, 
J3);
```

$$Answer := \ln\left(\frac{\sqrt{4+x^2}}{2}+\frac{x}{2}\right)+C$$

7. Calculate $\int \frac{3x+5}{x(x+1)}\,dx$ **.**

Solution:

```
> J := Int((3*x+5)/x/(x+1),x);
```

$$J := \int \frac{3x+5}{x(x+1)}\,dx$$

```
> R := student[integrand](J);
```

$$R := \frac{3x+5}{x(x+1)}$$

```
> R1 := convert(R, parfrac, x);
```

$$R1 := -\frac{2}{x+1} + \frac{5}{x}$$

```
> J1 := Int(R1, x) + C;
```

$$J1 := \int -\frac{2}{x+1} + \frac{5}{x}\,dx + C$$

```
> Answer := value(J1);
```

$$Answer := -2\ln(x+1) + 5\ln(x) + C$$

8. Calculate $\int \frac{5x^2+x+2}{x(x^2+1)}\,dx$ **.**

Solution:

```
> J := Int((5*x^2+x+2)/x/(x^2+1),x);
```

$$J := \int \frac{5x^2 + x + 2}{x(x^2+1)}\, dx$$

```
> R := student[integrand](J);
```

$$R := \frac{5x^2 + x + 2}{x(x^2+1)}$$

```
> R1 := convert(R, parfrac, x);
```

$$R1 := \frac{1+3x}{x^2+1} + \frac{2}{x}$$

```
> J1 := Int(R1, x) + C;
```

$$J1 := \int \frac{1+3x}{x^2+1} + \frac{2}{x}\, dx + C$$

```
> Answer := value(J1);
```

$$Answer := \frac{3}{2}\ln(x^2+1) + \arctan(x) + 2\ln(x) + C$$

9. The region in the first quadrant that is bounded above by $y = \sqrt{\dfrac{5x}{x^2+6}}$, below by $y = 0$, and on the right by $x = 2$ is rotated about the x-axis. What is the volume of the resulting solid of revolution?

Solution:

```
> Volume := Pi*Int(sqrt(5*x/(x^2+6))^2,x=0..2);
```

$$Volume := \pi \int_0^2 \frac{5x}{x^2+6}\, dx$$

```
> Answer := value(Volume);
```

$$Answer := \pi\left(\frac{5}{2}\ln(5) - \frac{5}{2}\ln(3)\right)$$

10. The region in the first quadrant that is bounded above by $y = \dfrac{4\arcsin(x)}{\pi}$ and below by $y = 2x$ for $0 \le x \le 1$ is rotated about the

y-axis. Express the volume of the resulting solid of revolution as an integral.

Solution:

```
> f := x -> 2*x; g := x -> 4*arcsin(x)/Pi;
```

$$f := x \rightarrow 2\,x$$

$$g := x \rightarrow \frac{4 \arcsin(x)}{\pi}$$

```
> plot([f(x), g(x)], x=0..1, color = [navy,blue], thickness=2);
  #f in navy,  g in blue
```




```
> cone := plot3d([r*cos(t),r*sin(t),f(r)], r = 0 ..1, t = Pi/6 ..
  11*Pi/6, style=WIREFRAME, color = PLUM):
> surface := plot3d([r*cos(t),r*sin(t),g(r)], r = 0 ..1, t = Pi/6
  .. 11*Pi/6,style=WIREFRAME, color = AQUAMARINE):
> plots[display](cone, surface, orientation=[10,75]);
```



```
> cylindrical_shell := plot3d([cos(t)/2,sin(t)/2,z], t = 0 ..
  2*Pi, z = g(1/2) .. f(1/2), style = PATCHNOGRID):
> plots[display](cone, surface, cylindrical_shell,
  orientation=[10,80]);
```



```
> radius := x; height := f(x) - g(x);
```

$$radius := x$$

$$height := 2\,x - \frac{4\,\arcsin(x)}{\pi}$$

```
> Volume_shells := Int(2*Pi*radius*height, x = 0 .. 1);  #Method
  of Cylindrical Shells
```

$$Volume\_shells := \int_0^1 2\,\pi\,x\left(2\,x - \frac{4\,\arcsin(x)}{\pi}\right)dx$$

```
> z1 := f(1/2); delta := 1/20; z2 := f(1/2) + delta; r0 :=
```

```
solve(g(x) = z1,x);
```

$$z1 := 1$$

$$\delta := \frac{1}{20}$$

$$z2 := \frac{21}{20}$$

$$r0 := \frac{\sqrt{2}}{2}$$

```
> washer_top := plot3d([r*cos(t),r*sin(t), z2], t = 0 .. 2*Pi, r
  = 1/2 .. r0, style = PATCHNOGRID, color = COLOR(RGB,0.85, 0.7,
  0.7)):
> washer_side := plot3d([r0*cos(t),r0*sin(t), z], t = 0 .. 2*Pi,
  z = z1 .. z2, style = PATCHNOGRID):
> plots[display](cone, surface, washer_top, washer_side,
  orientation=[10,80]);
```



```
> solve(y = g(x), x); solve(y = f(x), x);
```

$$\sin\left(\frac{y\,\pi}{4}\right)$$

$$\frac{y}{2}$$

```
> Volume_washers := Pi*Int( (sin(1/4*y*Pi))^2 - (y/2)^2, y=0..2);
  #Method of Washers
```

$$Volume\_washers := \pi \int_0^2 \sin\left(\frac{y\,\pi}{4}\right)^2 - \frac{y^2}{4}\, dy$$

```
> testeq(value(Volume_shells) = value(Volume_washers));
  #Verification that both methods give same
  answer
```

$$true$$

11. The region that is bounded above by the parabola $y = \dfrac{7-(x-3)^2}{3}$ and below by $y = \sqrt{x}$ for $1 \le x \le 4$ is rotated about the vertical line $x = 5$. *Using the Method of Cylindrical Shells*, express the volume of the resulting solid of revolution as an integral.

Solution:

```
> f := x -> (7-(x-3)^2)/3; g := x -> sqrt(x);
```

$$f := x \to \frac{7}{3} - \frac{1}{3}(x-3)^2$$

$$g := \mathrm{sqrt}$$

```
> plot([f(x), g(x)], x = 1 .. 4, color = [navy,blue],
  thickness=2, view = [0..4,0..2.5]);  #f in navy,  g in blue
```




```
> height := f(x) - g(x);  radius := 5 - x;
```

$$height := \frac{7}{3} - \frac{(x-3)^2}{3} - \sqrt{x}$$

$$radius := 5 - x$$

```
> Volume := 2*Pi*Int(radius*height, x = 1 .. 4);
```

$$Volume := 2\,\pi \int_{1}^{4} (5-x)\left(\frac{7}{3} - \frac{(x-3)^2}{3} - \sqrt{x}\right) dx$$

12. When stretching a spring from equilibrium to 40 cm beyond equilibrium, 64 J of work must be expended. From that point the force continues to increase until it reaches 416 N. How many centimeters beyond equilibrium is the spring stretched ?

Solution:

```
> eqn1 := Work = int( k*x, x = 0 .. 0.4);
```

$$eqn1 := Work = 0.08000000000\, k$$

```
> eqn2 := subs(Work = 64, eqn1);
```

$$eqn2 := 64 = 0.08000000000\, k$$

```
> eqn3 := k = solve(eqn2, k);
```

$$eqn3 := k = 800.$$

```
> eqn4 := F = k*x;
```

$$eqn4 := F = k\, x$$

```
> eqn5 := subs({F=416, eqn3}, eqn4);
```

$$eqn5 := 416 = 800.\, x$$

```
> Answer1 := solve(eqn5, x);  #answer in meters
```

$$Answer1 := 0.5200000000$$

```
> Answer2 := Answer1*100;  #answer in centimeters
```

$$Answer2 := 52.00000000$$

13. The shape of a tank is a parabola rotated about its axis of symmetry as shown in the figure below.



Initially the tank contains water that is 8 m deep. After pumping some of the water to the top of the tank, the remaining water is 2 m deep.

Express as an integral the amount of work (measured in Joules) that has been done.

Solution:

```
> solve( y = 9 - x^2, x);
```

$$\sqrt{-y+9}, -\sqrt{-y+9}$$

```
> Work = 9.81*Pi*10^3*Int( y*((-y+9)^(1/2))^2, y = 1 .. 7);
```

$$Work = 9810.00\,\pi \int_{1}^{7} y\,(-y+9)\, dy$$

14. Forty feet of a uniform cable hang over the side of a building. The cable weighs 3 lbs/ft. A 36 pound hook is attached to the dangling end of the cable. The cable is pulled up 12 feet before the hook snags on a flag pole. How many foot-pounds of work have been done?

Solution:

```
> L := 40*ft:
> Work = (36*lb)*(12*ft) + (L-12*ft)*(3*lb/ft)*(12*ft) +
  int(y*(3*lb/ft), y = 0*ft .. 12*ft);
```

$$Work = 1656\ lb\ ft$$